


## Introduksjon

Me skal lage vår eigen versjon av spelet **Flappy Bird**. Du styrar fuglen **Flakse** ved å trykke mellomromtasten for å flakse med vengjene. Du må halde Flakse flygande og prøve å styre mellom røra.



## Steg 1: Få Flakse til å dette

Me startar enkelt, med å introdusere Flakse og få han til å dette.



# Sjekkliste

- Start eit nytt Scratch-prosjekt. Slett katten ved å klikke på søppelkassen.
- Byt ut bakgrunnen med eit landskap. Desert er eit bra val.
- Legg til Flakse-figuren. Du treng ein figur med drakter for vengjer opp og vengjer ned. Til dømes kan du bruke Parrot.

- Byt namn på figuren til Flakse.
- Gi Flakse det følgjande skriptet:

```
når @greenFlag vert trykt på
gå til x: (-50) y: (0)
for alltid
   endra y med (-3)
slutt
```



## **Test prosjektet**

#### Klikk det grøne flagget.

• Startar Flakse midt på skjermen og beveger seg nedover?

### Steg 2: Få Flakse til å flyge

No vil me at Flakse skal flyge oppover når du trykkar mellomrom.



## **Sjekkliste**

- Clikk på Drakter og gi dei to draktene namna Venger opp og Venger ned.
- Gå tilbake til skript og legg til dette skriptet:

```
Når [mellomrom v] vert trykt
byt drakt til [Venger ned v]
gjenta (10) gongar
    endra y med (6)
slutt
byt drakt til [Venger opp v]
gjenta (10) gongar
    endra y med (6)
slutt
```



# Test prosjektet

#### Klikk det grøne flagget.

- Klarar du å kontrollere Flakse med mellomromtasten?
- La du merke til at om du trykkar mellomrom fleire gonger raskt etter kvarandre, så flaksar Flakse nokre gonger berre ein gong? Det er det neste me skal fikse.

## Steg 3: Gjer kontrollen betre

Me vil at Flakse skal reagere kvar einaste gong me trykkar mellomrom. Men når me trykkar mellomrom startar to løkker etter kvarandre. Viss me trykkar mellomrom att før desse to løkkene er ferdige, så skjer det ikkje noko. For å løyse problemet lagar me ein variabel som teller kor mange flaks Flakse skal gjere.



## **Sjekkliste**

- Lag ein ny variabel og kall den flaks. Vel at den berre skal gjelde For denne figuren. Klikk ok. Skjul variabelen ved å ta bort avhukinga framfor variabelen.
- Me skal gjere om på skriptet som startar med når mellomrom vert trykt. Dra i byt drakt til venger ned-klossen slik at den og klossane under losnar. Legg desse klossane til sides. Me skal bruke dei att veldig snart.
- Lag eit nytt skript. Legg merke til at klossane du la til sides kan bli brukt inne i gjenta til flaks = 0-løkka:

```
når @greenFlag vert trykt på
set [flaks v] til [0]
byt drakt til [Venger opp v]
for alltid
    gjenta til <(flaks) = [0]>
        endra [flaks v] med (-1)
       byt drakt til [Venger ned v]
        gjenta (10) gongar
            endra y med (6)
        slutt
        byt drakt til [Venger opp v]
        gjenta (10) gongar
           endra y med (6)
        slutt
    slutt
slutt
```

• Til slutt må du leggje det følgande skriptet på når mellomrom vert trykt-klossen som vart til overs tidlegare:

```
når [mellomrom v] vert trykt
endra [flaks v] med (1)
```

No skal det vere tre ulike skript på Flakse.



## Test prosjektet

#### Klikk det grøne flagget.

Flaksar Flakse ein gong for kvar gong du har trykka mellomromtasten?


## Steg 4: Legg til røyra

Me vil leggje til hindringar Flakse kan flyge forbi.





- 🕨 🔲 Legg til ein ny figur ved å klikke på *Teikn-*knappen,
- Uiss det står Punktgrafikk nede til høgre, må du klikke på Byt til vektorgrafikk-knappen.
- Klikk på zoom slik at du kan sjå heile teikneområdet.
- 🔲 Klikk på Rektangel, vel ei farge ved å klikke på Fyll-knappen øvst til venstre.
- Klikk og dra to boksar, ein frå toppen og ein frå botnen, i midten av teikneområdet. Det skal sjå omlag slik ut:



- Du kan skuggeleggje røyra ved å klikke på Fyll farge og velje ein av skuggemetodane nede til venstre. Vel to variantar av same farge, ein for framgrunnen og ein for bakgrunnen. Når du klikkar på ein firkant med fylleverktøyet får du ein fin effekt.
- Gi den nye figuren namnet Røyr.

## Steg 5: Få røyra til å bevege seg

No skal me få røyra til å flytte seg og gjere høgda tilfeldig, slik at me får ei hinderløype for Flakse.



- Klikk på Røyr-figuren og vel skript.
- Legg til desse to skripta:

```
når @greenFlag vert trykt på
gøym
set storleik til (200)%
for alltid
    lag klon av [meg v]
    vent (2) sekund
slutt

når eg startar som klon
gå til x: (240) y: (tilfeldig tal frå (-80) til (80))
vis
gjenta (120) gongar
    endra x med (-4)
slutt
slett denne klonen
```



## Test prosjektet

#### Klikk det grøne flagget.

- Har røyra opningar Flakse kan flyge gjennom?
- Uiss du synest det er vanskeleg å flyge Flakse gjennom opningane kan du til dømes endre på opninga mellom røyra med teikneverktyet. Ein annan moglegheit er å lage Flakse mindre.

## Steg 6: Finn ut om Flakse krasjar med røyra

For at spelet skal bli vanskeleg må spelaren styre Flakse gjennom åpningane mellom røyra utan å kome borti verken røyr eller kanten av skjermen. Me skal leggje til nokre klossar som merkar om Flakse krasjar.



## **Sjekkliste**

- Me legg til ein lyd som me kan spele av når Flakse krasjar. Klikk på Flakse-figuren og så på Lydar.
- Klikk på Vel lyd frå biblioteket.
- Vel ein krasjelyd for Flakse. Screech er ein kul lyd.

- Klikk deg tilbake til skript-fana.
- Legg til dette skriptet på Flakse:

```
når @greenFlag vert trykt på
vent til <<rører [kant v]?> eller <rører [Røyr v]?>>
start lyden [screech v]
sei [Du tapte!]
send meldinga [Tap v]
stopp [andre skript i figuren v] :: control
```

Så må du klikke på Røyr-figuren og leggje til dette skriptet:

```
når eg får meldinga [Tap v]
stopp [andre skript i figuren v] :: control
```



## Test prosjektet

#### Klikk det grøne flagget.

Stoppar spelet viss Flakse kjem borti eit røyr eller kanten av brettet?

## Steg 7: Legg til poeng

Spelaren skal få eitt poeng kvar gong Flakse flyg gjennom ei røyropning.



### **Sjekkliste**

- Me legg til ein lyd som kan spelast kvar gong Flakse får poeng. Klikk på Røyr-figuren og legg til ein lyd. Bird er eit lurt val.
- Gå tilbake til skript-fana.
- Lag ein ny variabel som skal gjelde For alle figurer. Kall den poeng.
- Legg til eit skript som set poenga til 0 når det grøne flagget blir klikka. Dette klarar du sjølv!
- Legg til dette skriptet på Røyr:

```
når eg startar som klon
vent til <(x-posisjon) < ([x-posisjon v] av [Flakse v])>
endra [poeng v] med (1)
start lyden [bird v]
```



## **Test prosjektet**

#### Klikk det grøne flagget.

• Får du poeng når Flakse flyg gjennom ei opning mellom røyra?

Korleis kan du gjere spelet lettare eller vanskelegare?



## Lagre prosjektet ditt

Supert, du har laga din eigen enkle versjon av Flappy Bird-spelet!

Her er nokre fleire ting du kan prøve:

### **Utfordring 1: Legg til tyngdekraft**

Når noko dett på grunn av tyngdekrafta aukar farta jo lengre fallet varar. Me skal prøve å etterlikne denne måten å dette på.

- Legg til ein ny variabel for Flakse som heiter løft. Variabelen skal gjelde for For denne figuren.
- Endre skriptet som styrer korleis Flakse dett:

```
når @greenFlag vert trykt på
set [løft v] til [0]
gå til x: (-50) y: (0)
for alltid
    endra y med (løft)
    endra [løft v] med (-0.2)
slutt
```

• Det neste steget er å endre flakseskriptet:

```
når @greenFlag vert trykt på
set [flaks v] til [0]
byt drakt til [Venger opp v]
for alltid
    gjenta til <(flaks) = [0]>
        endra [flaks v] med (-1)
        byt drakt til [Venger ned v]
        endra [løft v] med (4)
        vent (0.1) sekund
        byt drakt til [Venger opp v]
        vent (0.1) sekund
        slutt
slutt
```



### Test prosjektet

#### Klikk det grøne flagget.

Fell Flakse fortare jo lengre han dett?

## Utfordring 2: Fall ut av skjermen

Når spelaren tapar vil me at Flakse skal dette ned og ut av skjermen.

- Bytt ut send meldinga Tap-klossen med send meldinga Fall i skriptet som merkar når Flakse krasjar i kanten eller i eit røyr. Slett stopp-klossen på slutten av skriptet.
- Legg til desse nye skripta på Flakse-figuren:

```
når eg får meldinga [Fall v]
for alltid
    snu @turnRight (5) gradar
slutt

når eg får meldinga [Fall v]
gjenta til <(y-posisjon) < [-180]>
    endra y med (løft)
    endra [løft v] med (-0.2)
slutt
gøym
send meldinga [Tap v]
stopp [andre skript i figuren v] :: control
```

• Du må leggje til ein vis-kloss og setje retninga for Flakse når spelet startar på nytt.



# Test prosjektet

#### Klikk det grøne flagget.

- Fell Flakse ut av skjermen når han treff eit røyr?
- Flyg Flakse riktig veg når spelet startar att?

## **Utfordring 3: Legg til rekordar**


- Lag ein ny variabel og merk av Nett-variabel (lagra på nett)-boksen. Kall variabelen Rekord.
- Når spelet er ferdig må me sjekke om det er ein ny rekord. Det gjer me ved å endre Tapskriptet på røyr slik:

```
når eg får meldinga [Tap v]
viss <(poeng) > (Rekord)>
    set [Rekord v] til (poeng)
slutt
stopp [andre skript i figuren v] :: control
```

#### Nettvariablar

Nettvariablar gir deg moglegheita til å hugse ting mellom ulike brukarar og ulike køyringar av spelet ditt (i motsetnad til vanlege variablar der verdiane blir nullstilt mellom kvar gong du trykkjer det grøne flagget). Dette er akkurat det me treng for å hugse kva rekorden i spelet er.

Nettvariablar er framleis eksperimentelle, og for tida får ikkje nye Scratch- bruarar lov til å lage slike variablar. Viss du ikkje får lov til det må du hoppe over den siste utfordringa.



#### Klikk det grøne flagget.

• Oppdaterer rekorden seg kvar gong du set ein ny rekord?



#### Gratulerer, du er ferdig med spelet! Kva er rekorden din?

Ikkje gløym å dele spelet med venene dine. Trykk på Legg ut for at andre skal få prøve!

Lisens: Code Club World Limited Terms of Service