

○ Felix og Herbert



PÅ BOKMÅL



LAST NED PDF

Introduksjon

Me skal laga eit spel der katten **Felix** skal fanga musa **Herbert**. Du styrer Herbert med musepeikaren og skal prøva å unngå å bli tatt av Felix. Jo lenger du unngår han jo fleire poeng får du, men blir du tatt, går poengsummen din ned.



Steg 1: Felix følgjer musepeikaren

Me ynskjer at katten Felix skal følge etter musepeikaren.




Sjekkliste

- ☐ Start eit nytt prosjekt.



- ☐ Trykk på **i** i hjørnet av **Sprite1** og byt namn på figuren til `Felix`.

- ☐ Sørg for at Felix kun ser til høyre og venstre ved å setje rotasjonsmåte til



- ☐ Klikk på scenen ved sida av Felix i vinduet for figurar. Vel fana *Bakgrunnar* og trykk på for å importera ein ferdig bakgrunn. Vel den bakgrunnen du vil.
- ☐ Klikk på Felix, vel *Skript-fana* og lag dette skriptet:

```
Når @greenFlag vert trykt på
for alltid
  peik mot [musepeikar v]
  gå (10) steg
  neste drakt
  trommeslag (3 v) som varer (0.25) takter
slutt
```



Test prosjektet

Klikk på det grønne flaget.

- ☐ Følgjer Felix musepeikaren?
- ☐ Ser det ut som han går når han bevegar seg?
- ☐ Bevegar han seg med rett hastighet?
- ☐ Klikk det raude stopp-symbolet for at Felix skal slutta å følgje etter musepeikaren.



Lagre prosjektet

Om du er pålogga med din eigen scratchbrukar lagrar Scratch alle prosjekta dine automatisk med jevne mellomrom. Det kan allikevel vera lurt å lagra manuelt innimellom.

- ☐ I filmenyen, vel *Lagre no.*

Viss du ikkje har brukar kan du ikkje lagre, berre fortsett til steg 2.

Steg 2: Felix jagar Herbert

Nå ynskjer me at Felix skal jaga musa Herbert istadenfor musepeikaren.



Sjekkliste



- ☐ Lag ein ny figur ved å trykke på og vel figuren `Dyr/Mouse1`.
- ☐ Byt namn på figuren til `Herbert`.
- ☐ Gjer Herbert mindre enn Felix ved å trykke på (øvtst mot midten av vindauga). Prøv seks klikk.
- ☐ Gje Herbert dette skriptet:

```
Når @greenFlag vert trykt på
for alltid
    gå til [musepeikar v]
    peik mot [Felix v]
slutt
```



Test prosjektet

Klikk på det grønne flaget.

- ☐ Flyttar Herbert seg med musepeikaren?
- ☐ Jagar Felix Herbert?

Steg 3: Felix seier når han har fanga Herbert

Me vil at Felix skal veta når han har fanga Herbert og fortelje det til oss.



Sjekkliste

- ☐ Endre skriptet til Felix til dette:

```
Når @greenFlag vert trykt på
for alltid
    peik mot [musepeikar v]
    gå (10) steg
    neste drakt
    trommeslag (3 v) som varer (0.25) takter
```

```
viss <rører [Herbert v]?>
    sei [Tok deg!] i (1) sekund
slutt
slutt
```



Test prosjektet

Klikk på det grønne flaget.

- ☐ Seier Felix frå når han har fanga Herbert?

Steg 4: Herbert blir eit spøkelse når han vert fanga

I tillegg til at Felix seier noko, vil me nå at Herbert blir forvandla til eit spøkelse når han vert fanga.





Sjekkliste

- ☐ Endra skriptet til Felix slik at det sender ein melding og lagar ein lyd når han fangar Herbert:

```
Når @greenFlag vert trykt på
for alltid
    peik mot [musepeikar v]
    gå (10) steg
    neste drakt
    trommeslag (3 v) som varer (0.25) takter
    viss <rører [Herbert v]?>
        send meldinga [Fanga!]
        trommeslag (1 v) som varer (0.25) takter
        sei [Tok deg!] i (1) sekund
        vent (1) sekund
    slutt
slutt
```

- ☐ Vel Herbert og gå til Drakter-fana.



- ☐ Hent ein ny drakt ved å trykke på  og vel Fantasi/ghost2-a
- ☐ Gjer drakta mindre ved å velge  og trykke seks gonger på spøkelsesdrakta.
- ☐ Endra namna på Herberts draktar slik at musedrakta heiter levande og spøkelsesdrakta heiter daud.
- ☐ Gå til skript-fana, og lag eit nytt skript for Herbert for å gjera han om til eit spøkelse. Ikkje slett det gamle skriptet:

```
når eg får meldinga [Fanga! v]
byt drakt til [daud v]
vent (0.5) sekund
byt drakt til [levande v]
```



Test prosjektet

Klikk på det grønne flaget.

- ☐ Forvandlas Herbert til eit spøkelse når han vert fanga?
- ☐ Spelar Felix dei rette lydane til rett tid?
- ☐ Står Felix stille lenge nok til at Herbert kjem seg unna?

Steg 5: Telje poeng

La oss legge til ein poengsum slik at me kan sjå kor flink ein er til å halde Herbert i live. Me startar med poengsummen null og auker den med ein for kvart sekund. Hvis Felix fangar Herbert, minker me poengsummen med ti.



Sjekkliste

- ☐ På Skript-fana under kategorien Data, lag ein ny variabel. Kall variabelen for Poeng, og la den gjelde for alle figurar.

Ny variabel

Nytt variabelnavn:

Poeng

☒ For alle figurer ☐ For denne figuren

Avbryt OK

Legg merke til at variabelen Poeng dukka opp øvst til venstre i spelet ditt.

- ☐ Klikk på Scene til venstre på skjermen, ved sida av Figurar. Lag disse to skripta på scenen:

```
Når @greenFlag vert trykt på
sett [Poeng v] til (0)
for alltid
  vent (1) sekund
  endra [Poeng v] med (1)
når eg får meldinga [Fanga! v]
  endra [Poeng v] med (-10)
```



Test prosjektet

Klikk på det grønne flaget.

- ☐ Aukar poengsummen med ein kvart sekund?
- ☐ Går poengsummen ned med ti når Herbert vert fanga?
- ☐ Kva skjer dersom Herbert vert fanga før du har ti poeng?
- ☐ Går poengsummen tilbake til null når du starter spelet på nytt?



Lagre prosjektet

Du er ferdig. Godt gjort. Nå kan du spele spelet!

Viss du er logga inn med din scratchbrukar kan du dele spelet med familie og venner ved å trykke [Legg ut](#) på menylinjen.

Lisens: [Code Club World Limited Terms of Service](#)