

# Introduksjon

I denne oppgåva skal me lage eit spel der det er om å gjere å jage bort dei slemme gutane ved å kaste snøball på dei. Undervegs lærer me korleis me kan gjere spelet meir utfordrande etter kvart ved at figurane me speler mot blir raskare og vanskelegare å jage bort.



# Steg 1: Ein snøballkastar

Me startar med å lage ein passande bakgrunn og snøballkastaren me skal styre.




• Start eit nytt prosjekt. Slett kattefiguren.

• Me skal lage ein bakgrunn. Til dette spelet treng me ein ganske enkel bakgrunn med litt vinterstemning. Den kan me teikne sjølv:

Vel for å teikne ein ny bakgrunn. Pass på at du brukar vektorgrafikk. Teikn ein stor firkant som dekker heilt bakgrunnen. Vel ei mørkeblå framgrunnsfarge og ei lysare bakgrunnsfarge. Klikk på målingsspannet for å fylle firkanten med farge, og vel ein overgang nedst til venstre. Klikk på bakgrunnen for å fylle den med farge.

Så vel du kvit som framgrunnsfarge og lagar ein mindre boks som dekker nedre del av bakgrunnen. Fyll den ed passande fargar.



Kall bakgrunnen for spel.

Legg det følgjande skriptet på scena:

```
når @greenFlag vert trykt på
send meldinga [start v]
```

Dette trikset har du kanskje sett før. Det gjer det enklare for oss å kome attende og leggje til ein startmeny og liknande.

- Neste steg er å finne ein passande figur som me kan styre rundt og kaste snøballar med. Me har brukt snømannen Fantasi/Snowman, men du kan bruke ein annan figur du likar. Gi figuren namnet Helten.
- Lag ein ny variabel, hastigheit, som berre gjeld for denne figuren, og lag dette skriptet:

```
når @greenFlag vert trykt på
set storleik til (75) %
avgrens rotering til [venstre-høgre v]
set [hastigheit v] til [5]
```

På det grøne flagget legg me innstillingar som skal gjelde for Helten gjennom heile spelet. Du må gjerne endre på storleiken og hastigheita etter kvart som du testar spelet, slik at du finn

innstillingar du likar.

- No kan me lage hovudløkka til Helten. Lag ei løkke som startar på
- meldinga start:

```
når eg får meldinga [start v] gå til x: (0) y: (-75) for alltid slutt
```

- Inne i løkka treng me to viss-klossar som flyttar Helten hastigheit steg mot høgre eller venstre når piltastane høgre eller venstre blir trykka på. Desse lagar du sjølv.
- 🔲 Til slutt lagar du ein viss-kloss som sender meldinga kast når mellomrom-tasten blir trykka på.



# Test prosjektet

### Klikk på det grøne flagget.

- Beveger figuren din seg rundt på scena når du trykkar høgre og venstre piltast? Førebels skal det ikkje skje noko når du trykkar mellomrom, sidan det ikkje er nokon figurar som svarar på kastmeldinga.
- Sannsynlegvis må du endre litt på y-posisjonen figuren din startar på, slik at det passar med bakgrunnen.
- Uss du har ein figur med fleire drakter kan du bruke neste drakt-klossar for å animere figuren din slik at den beveger seg meir naturleg.

### Steg 2: Mange snøballar

Det blir ingen snøballkrig utan snøballar. No skal me lage kjempemange av dei!



## **Sjekkliste**

- Teikn ein ny figur. Bruk sirkelverktøyet til å teikne ein liten kvit sirkel, og fargelegg den slik at den blir heilt kvit. Trykk på og pass på at senterpunktet er mitt på snøballen. Kall figuren snøball.
- På same måte som for Helten set me opp nokre standardinnstillingar for snøballen. Lag en variabel hastigheit som berre gjeld for denne figuren.

```
når @greenFlag vert trykt på
gøym
set storleik til (40) %
set [hastigheit v] til [10]
```

• Me brukar gøym fordi me vil lage kloner (kopiar) av snøballen som me kastar av garde. Dette er eit veldig nyttig triks i Scratch. Sjølve koden som reagerer på kast-meldinga er veldig enkel.

```
når eg får meldinga [kast v]
lag klon av [meg v]
```

• Sjølve oppførselen til kvar enkelt snøball programmerer me på ein når eg startar som klonkloss.

```
når eg startar som klon
gå til [Helten v]
peik i retning ([retning v] av [Helten v])
vis
gjenta til <rører [kant v]>
    gå (hastigheit) steg
slutt
slett denne klonen
```



## Test prosjektet

### Klikk på det grøne flagget.

- Kva skjer når du trykkar mellomrom-tasten? Blir det kasta snøballar i rett retning?
- Det ser kanskje ikkje ut som om Helten kastar snøballane med hendene? Du kan leggje på nokre Bevegelse-klossar rett før du viser snøballen for at det skal sjå betre ut.
- Som ei lita utfordring kan du prøve å leggje på litt effekt av tyngdekrafta på snøballen ved å endre litt på y medan snøballen flyr.
- Eit problem me kan leggje merke til er at Helten er for flink til å kaste snøball! Kvar gong me trykkar mellomrom blir det kasta mange snøballar. Dette kan me løyse enkelt ved å vente til mellomromtasten blir sluppe før me kastar snøballen. Legg til

```
vent til <ikkje <tasten [mellomrom v] er trykt?>>
```

på Helten-figuren før kast-meldinga blir sendt.

### Steg 3: Slemme gutar

No skal me lage slemme gutar som prøver å ta oss. Me skal jage dei bort med snøballar.



# Sjekkliste

- Lag ein ny figur. Me har brukt Mennesker/Boy3 walking, men du kan bruke den figuren du likar. Kanskje er det meir morosamt å ha snøballkrig mot eit monster? Kall figuren skumling.
- På same måte som for snøballane vil me lage kloner av skumling. Men fyrst bestemmer me korleis han skal sjå ut. Lag ein ny variabel hastigheit som berre gjeld for denne figuren. Så lagar du skriptet

```
når @greenFlag vert trykt på
gøym
gå til x: (0) y: (-70)
avgrens rotering til [venstre-høgre v]
```

```
set storleik til (30) %
set [hastigheit v] til [3]
```

• Me vil at klonene skal dukke opp med litt tilfeldig mellomrom, og frå begge sider av skjermen.

```
når eg får meldinga [start v]
for alltid
  viss <(tilfeldig tal frå (0) til (1)) = [0]>
     peik i retning (90 v)
     set x til (-250)
ellers
     peik i retning (-90 v)
     set x til (250)
slutt
  lag klon av [meg v]
  vent (tilfeldig tal frå (2) til (4)) sekund
slutt
```

• På same måte som for snøballane må me bestemme oppførselen for kvar enkelt skumling.

Lag ein når eg startar som klon-kloss der du fyrst viser figuren, og så lagar ei for alltid-løkke der figuren beveger seg hastigheit steg og så ventar ein augneblink, til dømes 0,1 sekund.

• Så lagar me ein *ny* når eg startar som klon-kloss der me undersøker om me treffer anten ein snøball eller Helten.

```
når eg startar som klon
for alltid
    viss <rører [Helten v]?>
        send meldinga [slutt v]
        slett denne klonen
    slutt
    viss <rører [Snøball v]?>
        slett denne klonen
    slutt
```

Grunnen til at desse må liggje i eit eige skript er at det fyrste skriptet ventar litt mellom kvar gong figuren tek eit steg. Hadde me lagt desse viss-klossane i det same skriptet ville me berre sjekka om skumling vart treft av ein snøball mellom ventinga. Ved å lage eit eige skript sjekkar me det heile tida.



### Test prosjektet

#### Klikk på det grøne flagget.

- Dukkar det opp skumlingar? Kjem dei frå begge sider? Går dei mot midten?
- Wva skjer når ein <code>skumling</code> blir treft av ein <code>snøball</code>? Du kan kanskje leggje på ein passande lydeffekt og animasjon før klonen blir sletta? Me ser og at snøballen flyr vidare etter at den har truffe ein <code>skumling</code>. Prøv å endre i skriptet for <code>snøball</code> slik at snøballen forsvinn når den treff.
- Wva skjer når ein skumling tek Helten?

### Steg 4: Telje poeng og avslutte spelet

No som me nesten har eit ferdig spel, så vil me telje poeng!



# Sjekkliste

- A telje poeng er ganske lett. Lag ein variabel som heiter Poeng og la den gjelde for alle figurar. La variabelen vere synleg på scena, slik at me ser kor mange poeng me har fått.
- Pass på at Poeng blir sett til 0 når meldinga start blir sendt, til dømes med eit skript på scena.
- Endre Poeng med 1 når ein Skumling blir treft av ein snøball.

Tidlegare laga me meldinga slutt som blir sendt ut når Helten blir teken av ein skumling. No skal me bruke denne til å avslutte spelet. Fyrst lagar me ein meny og ein bakgrunn som fortel att me tapte.

• Klikk på scene til venstre for Figurer, og vel Bakgrunnar-fana. Lag to kopiar av bakgrunnen din og kall dei henhaldsvis Meny og slutt.

På Meny-bakgrunnen kan du lage ein fin tittel. Skriv Trykk 'S' for å starte.

På slutt-bakgrunnen kan du skrive ei passande melding for når spelet er slutt. Skriv Trykk 'S' for å spele igjen.

- Endre litt på skripta på scena. Fyrst vil me berre vise menyen når det
- · grøne flagget blir klikka på:

```
når @greenFlag vert trykt på
byt bakgrunn til [Meny v]
```

Så vil me starte spelet når s blir trykka på:

```
når [s v] vert trykt
send meldinga [start v]
```

Pass på at du byttar til bakgrunnen spel når meldingen start blir motteke, og til bakgrunnen slutt når meldinga slutt blir motteke.

• Til slutt må me passe på at spelet faktisk blir avslutta når slutt-meldingen blir sendt. Legg på skript for å slette denne klonen på Snøball og Skumling, og skript for å gøyme Helten når slutt blir motteke.



### Test prosjektet

### Klikk på det grøne flagget.

- Fungerer programflyten? Startar spelet når du trykkar s? Blir spelet avslutta når Helten blir fanga? Visast dei riktige bakgrunnane når dei skal?
- Du må gjerne leggje på lydeffektar og enkle animasjonar der det passar.

### Steg 5: Store, slemme gutar

Til slutt skal me gjere spelet vanskelegare di lengre me speler.

- Fyrst lagar me ein ny variabel Nivå som skal gjelde for alle figurar. Denne styrer kor vanskeleg spelet skal vere. Du må gjerne la den vere synleg.
- Me reknar ut nivå basert på Poeng. Utvid skriptet ditt som tek imot start-meldinga på scena med den følgjande løkka:

```
for alltid set [Nivå v] til ((1) + ([golv v] av ((Poeng) / (5)))) slutt
```

Funksjonen gulv rundar nedover. Så med denne løkka aukar Nivå for kvar femte skumling me jagar vekk.

• No kan me bruke Nivå til å endre skumling. Til dømes kan dei bli større, gå raskare og kanskje trenge fleire snøballar før dei blir borte.

Lag eit par nye variablar som gjeld for denne figuren: <code>slem</code> og <code>Liv</code>. <code>slem</code> seier kor stor og slem den enkelte <code>skumling</code> er. Ved å la det vere eit tal mellom 1 og <code>Nivå</code> blir spelet vanskelegare etter kvart som <code>Nivå</code> aukar. Til dømes kan du leggje til desse klossane i hovudskriptet til <code>skumling</code>:

```
når eg startar som klon
set [Slem v] til (tilfeldig tal frå (1) til (Nivå))
set [Liv v] til (Slem)
endra [hastigheit v] med (Slem)
endra [farge v]-effekt med ((10) * (Slem))
endra storleik med ((5) * (Slem))
vis
for alltid
    gå (hastigheit) steg
    vent (0.1) sekund
slutt
```

- For at ein <code>skumling</code> skal tåle fleire snøballar må me endre litt på kva som skjer når den blir treft. I staden for å berre slette klonen vil me endre <code>Liv</code> med -1, og så gi poeng og slette klonen berre viss <code>Liv</code> er 0.
- Kanskje bør me gi fleire poeng for å jage bort ein større skumling. Det gjer du ved å endre Poeng med slem når ein skumling blir jaga bort.
- Uiss me får meir enn eitt poeng for kvar skumling bør me endre på korleis me reknar ut Nivå. Viss ikkje blir spelet veldig fort mykje vanskelegare. Til dømes kan du bruke denne utrekninga:

```
for alltid set [Nivå v] til ((1) + ([gulv v] av ([kvadratrot v] av ((Poeng) / (3))))) slutt
```

Då aukar framleis nivået for omlag kvar femte skumling som blir jaga bort.



## **Test prosjektet**

Klikk på det grøne flagget.

- Blir spelet vanskelegare etter kvart? Prøv å endre på dei ulike variablane og innstillingane me har laga slik at spelet blir passe vanskeleg.
- 🔲 Har du fleire idear til korleis spelet kan bli endå meir morosamt? Prøv dei ut!

Lisens: CC BY-SA 4.0