

# △ Ørkenløp



PÅ BOKMÅL



LAST NED PDF

## Introduksjon

Dette er eit spel for to, der ei papegøye og ei løvinne kjempar om å kome fyrst gjennom ørkenen. Kvar spelar må trykke ein tast så fort og så ofte som mogleg for å flytte figuren sin, og den som kjem fyrst til kanten av skjermen vinn.






## Steg 1: Lag ei scene og legg til figurar

Me startar med å få på plass bakgrunnen og figurane.



### Sjekkliste

- ☐ Klikk på Scene og vel ein ferdig bakgrunn, . Vel `Natur/desert`.
- ☐ Fjern katten ved å høgreklikke på figuren og vel `slett`.
- ☐ Legg til ein ny figur ved å trykke på . Vel `Dyr/Lionness`.

- ☐ Legg til endå ein ny figur: vel `Dyr/Parrot`. Krymp figuren slik at den er omlag like stor som løvinna ved å bruke .

## Steg 2: La løvinna og papegøya bevege seg

Me vil at figurane skal bevege seg når du trykkar på ein knapp.



### Sjekkliste

- ☐ Vel løvefiguren og få den til å gå 4 steg når du trykkar `L`-tasten.  

```
når [l v] vert trykt  
gå (4) steg
```
- ☐ Vel papegøyefiguren og la den gå 4 steg når du trykkar `A`-tasten.  

```
når [a v] vert trykt  
gå (4) steg
```



### Test prosjektet

Klikk på det grønne flagget.

- ☐ Beveger løvinna og papegøya seg over skjermen når du trykkar på `A`- og `L`-tastane?

## Steg 3: Start kappløpet

No må me starte kappløpet og kåre ein vinnar! Me startar med å lage ein startknapp.



### Sjekkliste

- ☐ Legg til ein ny figur. Vel `Ting/Button3`. Flytt den til midten av scena.
- ☐ Klikk på `Drakter`-fana og verktøyet `T` for å leggje til tekst. Trykk på venstre kant av knappen for å leggje til eit tekstfelt og skriv inn teksten `Start`. Du kan flytte på teksten ved å trykkje på den ein gong, og du endrar innhaldet ved å dobbeltklikke.
- ☐ Legg til eit skript som viser figuren når spelet startar:  

```
når @greenFlag vert trykt på  
vis
```
- ☐ I tillegg vil me at når du klikkar på knappen skal den fyrst telje ned frå 3, seie `SPRING!` og så bli gøymt. Dette ordnar du med følgjande skript:  

```
når denne figuren vert trykt på  
sei [3] i (1) sekund
```

```
sei [2] i (1) sekund
sei [1] i (1) sekund
sei [SPRING!] i (1) sekund
gøym
```



## Test prosjektet

Klikk på det grønne flagget, og så på startknappen.

- ☐ Tel knappen ned?
- ☐ Seier den `SPRING!`?
- ☐ Blir den borte?

Me vil at figurane berre skal bevege seg etter at kappløpet har starta, og me vil vite når kappløpet er over.



## Sjekkliste

- ☐ For å vite når kappløpet har starta og slutta lagar me ein variabel med namnet `kappløp`. Variabelen skal vere tilgjengeleg for alle figurar. Fjern avhukinga framfor variabelen slik at den ikke visast på scena.
- ☐ Set `kappløp` til 0 når spelet startar ved å forandre når grønt flagg vert trykt på-skriptet slik:

```
når @greenFlag vert trykt på
vis
set [kappløp v] til [0]
```

- ☐ Når nedteljinga er ferdig og løpet startar forandrar me `kappløp`-verdien til 1. Dette gjer du ved å leggje til klossen Set `kappløp` til 1 under sei 1 i 1 sekund i skriptet som startar med når denne figuren vert trykt på.
- ☐ No må me lage ein regel som seier at figurane berre får lov til å bevege seg etter at løpet har starta – det vil seie når `kappløp` har verdien 1. Klikk på papegøya og endre:

```
når [a v] vert trykt
viss <(kappløp) = [1]>
  gå (4) steg
slutt
```

- ☐ Gjenta det same for løvinna.



## Test prosjektet

Klikk på det grønne flagget.

- ☐ Kan løvinna og papegøya berre flytte seg når nedteljinga er ferdig?

## Steg 4: Avslutte kappløpet

No vil me finne ut kven som vinn kappløpet, og i tillegg gjere klart for ein ny runde.



### Sjekkliste

- ☐ Legg til ein kloss i papegøya sitt skript som seier `set kappløp til 0` viss figuren er borti kanten av skjermen:

```
når [a v] vert trykt
viss <(kappløp) = [1]>
  gå (4) steg
  viss <rører [kant v]?>
    set [kappløp v] til [0]
  slutt
slutt
```

- ☐ Spel inn ein lyd som skal spelast av viss papegøya vinn.

Trykk på `Lyder`-fana og så mikrofon-ikonet og spel inn ein morosam lyd! Opptaket startar når du har klikka på disken slik at den blir raud. Klikk stopp (firkanten) når du er ferdig, og gi lyden eit namn - til dømes `Polly`. Nokre nettlesarar kan be deg om lov til å spele inn lyd. Viss du ikkje vil godta det kan du bruke lydane som følgjer med figurane.

- ☐ Så legg du til klossane som speler av lyden og let papegøya fortelje at den vann:

```
når [a v] vert trykt
viss <(kappløp) = [1]>
  gå (4) steg
  viss <rører [kant v]?>
    set [kappløp v] til [0]
    start lyden [Polly v]
    sei [Polly vann!] i (3) sekund
  slutt
slutt
```

- ☐ Gjer tilsvarande for løvinna.



### Test prosjektet

Klikk på det grøne flagget.

- ☐ Kan du trykkje på startknappen og så bevege dyra med tastane `A` og `L`?
- ☐ Kjem den riktige vinnarlyden og riktig melding opp på skjermen?

## Steg 5: Nullstill spelet

Når kappløpet er over må me fortelje dei andre figurane at spelet er over og nullstille spelet, slik at det er klart for ein ny runde.



## Sjekkliste

- ☐ Klikk på papegøyefiguren og legg til ein kloss som sender melding `Avslutt` etter at figuren seier den har vunne.

```
når [a v] vert trykt
viss <(kappløp) = [1]>
  gå (4) steg
  viss <rører [kant v]?>
    set [kappløp v] til [0]
    start lyden [Polly v]
    sei [Polly vann! v] i (3) sekund
    send meldinga [Avslutt v]
  slutt
slutt
```

- ☐ Me treng eit nytt skript som lyddar etter denne avslutningsmeldinga og flyttar papegøya attende til start.

```
når jeg mottar [Avslutt v]
set x til (-170)
```

- ☐ Gjer det same for løvinna. Test ulike `x`-verdiar for å vere sikker på at løvinna og papegøya startar frå same stad.

- ☐ For at figurane skal stå på startstreken når kappløpet startar den aller fyrste gongen må me leggje til følgjande klossar på begge figurane:

```
når @greenFlag vert trykt på
set x til (-170)
```

- ☐ For at spelarane skal kunne klikke i gang nye runder må me passe på at start-knappen kjem til syne att etter kvar avslutta runde. Klikk på startknapp-figuren og legg til eit skript som viser knappen når avslutningsmeldinga blir motteke.

```
når jeg mottar [Avslutt v]
vis
```



## Test prosjektet

Klikk på det grønne flagget.

- ☐ Kan du spele mot ein ven? Ein av dykk styrer papegøya ved å trykke `A`, og den andre løvinna ved å trykke `L`.



## Lagre prosjektet

Spelet er ferdig! Viss du vil kan du velje `Legg ut` slik at venene og familien din kan prøve det.

Under finn du nokre framlegg og idear til korleis du kan utvide spelet og gjere det meir interessant.

## Utfordring 1: Legg til ein rakett!

- ☐ **Legg til ein rakett** som kan brukast ein gong per kappløp og som flyttar papegøya eller løvinna **30 steg på ein gong**.
- ☐ **Legg til ei ny drakt** med ild som kjem ut bak figurane. La denne bli aktivert når raketten avfyrast.
- ☐ **Lag ein lyd** som figuren kan gi frå seg når raketten avfyrast.

Under er eit framlegg til korleis eit rakett-skript kan sjå ut. Du må leggje til nokre lydar og variablar på eiga hand.

```
når [q v] vert trykt
viss <<(kappløp) = [1]> og <(rakett_brukt) = [0]>>
  bytt drakt til [parrot-rakett v]
  set [rakett_brukt v] til [1]
  gå (30) steg
  start lyden [motorcycle passing v]
  viss <rører [kant v]?>
    set [kappløp v] til [0]
    start lyden [Polly v]
    sei [Polly vann! v] i (3) sekund
    send meldinga [Avslutt v]
  slutt
  bytt drakt til [parrot-a v]
slutt
```



## Test prosjektet

Klikk på det grønne flagget.

- ☐ Fungerer raketten slik du hadde planlagt?
- ☐ Kva skjer viss ein prøver å bruke raketten to gonger?
- ☐ Kva skjer viss ein brukar raketten for å kome heilt til mål?

## Utfordring 2: Bruk eigendefinerte klossar for å forenkle skriptet ditt

Koden som brukast for å sjekke om kappløpet er over brukast to stader for kvar figur: både når den beveger seg normalt og når den beveger seg med rakett. Me kan forenkle skriptet vårt ved å lage ein eigendefinert kloss. Dette er ei samling kode som brukast fleire stader. Det er nesten som å lage vår eigen Scratch-kodekloss!

- ☐ Vel papegøya sitt skript.
- ☐ Vel Flere klosser-paletten og klikk på Lag en kloss-knappen.
- ☐ Kall klossen din `ferdig` og trykk OK.
- ☐ No får du ein `definer ferdig-kloss` i skriptvindauget ditt. Flytt den litt for seg sjølv.

- ☐ Riv laus heile `vissrører kant?-klossen` og dra den til `definer ferdig-klossen`.
- ☐ Kan du dra `ferdig-klossen` frå paletten og bruke den på same måte som andre kodeklossar?
- ☐ Slett den andre `vissrører kant?-klossen` frå skriptet ditt og erstatt den med ein `ferdig-kloss`.

```
definer ferdig
viss <rører [kant v]?>
  set [kappløp v] til [0]
  start lyden [Polly v]
  sei [Polly vann! v] i (3) sekund
  send meldinga [Avslutt v]

når [a v] vert trykt
viss <(kappløp) = [1]>
  gå (4) steg
  ferdig

når [q v] vert trykt
viss <<(kappløp) = [1]> og <(rakett Brukt) = [0]>>
  bytt drakt til [parrot-rakett v]
  set [rakett Brukt v] til [1]
  gå (30) steg
  start lyden [motorcycle passing v]
  ferdig
  bytt drakt til [parrot-a v]
```

- ☐ Gjer dette koden din enklare å lese? Kan du lage ein tilsvarende eigendefinert kloss for løvinna?



## Test prosjektet

Klikk på det grønne flagget.

- Virkar spelet framleis slik det skal?
- Er spelet ferdig når papegøya eller løvinna kjem til kanten av skjermen?



## Lagre prosjektet

Veldig bra! No er du ferdig og kan kose deg med spelet du har laga!

Lisens: [Code Club World Limited Terms of Service](#)